

Pràctiques Integrades 1er de Matemàtiques

Curs 2003–04

Índex

VII Integració	3
21 Integrals de Riemann	3
21.1 La integral de Riemann	3
21.2 El mètode dels trapezis	5
21.3 Mètode de Simpson	6
21.4 La funció integral	8
21.5 Àrea entre dues gràfiques	8
22 Càlcul de primitives	9
22.1 Canvi de variable o substitució	10
22.2 Mètode d'integració per parts	11
22.3 Integrals racionals	13
VIII Diagonalització i formes de Jordan	17
23 Endomorfismes, vectors i valors propis, diagonalització...	17
23.1 Polinomi característic i valors propis	17
23.2 Vectors propis	17
24 Formes de Jordan	21
24.1 Potències de matrius.	22

Part VII

Integració

21 Integrals de Riemann

En aquesta secció analitzarem el concepte de la integral de Riemann. Donada una funció f amb valors reals definida en un interval $[a, b]$, la integral de Riemann dóna sentit i calcula l'àrea de la regió plana limitada pel gràfic de la funció, l'eix d'abscisses i les rectes $x = a$ i $x = b$. El mètode es basa en aproximar la regió mitjançant unions d'un nombre finit de figures per a les quals sabem calcular l'àrea: rectangles i trapezis per exemple.

Per a fer aquest anàlisi utilitzarem les funcions que proporciona el paquet **Student[Calculus1]** (que ja es hora que carregueu amb la comanda `with(Student[Calculus1]);`). Ens centrarem principalment en veure com es poden utilitzar les comandes `RiemannSum`, `ApproximateInt` i les opcions que tenen.

21.1 La integral de Riemann

La comanda de Maple `RiemannSum` és la que ens permet analitzar sumes de Riemann per a una funció f . En l'exemple següent veiem com fer una representació gràfica de la suma de Riemann donada $f(x)$ i un interval $[a, b]$ considerant com altures dels rectangles els valors de la funció f en el cantó dret de cada sub-interval. Per defecte parteix l'interval inicial en deu subintervalls d'igual longitud i si volem determinar aquest nombre nosaltres mateixos ho podem fer afegint la opció `partition=n`, on n és el nombre de particions que es vol fer.

Exemple 21.1

```
> f:=x->x^2;
> RiemannSum(f(x),x=0..1,method=right,output=plot);
> RiemannSum(f(x),x=0..1,method=right,output=plot,partition=5);
```

Noteu que en el gràfic que apareix surt el valor aproximat de la integral (el que surt si es fa `evalf` de la comanda `int` corresponent) i el valor de la suma de l'àrea dels rectangles que s'han dibuixat. Si el que volguéssim fos aquest valor de l'aproximació per a fer altres càlculs s'hauria de canviar l'opció `output=plot` per `output=value`.

Exemple 21.2

```
> f:=x->x^2;
> RiemannSum(f(x),x=0..1,method=right,output=value);
> RiemannSum(f(x),x=0..1,method=right,output=value,partition=5);
```

També es pot canviar la mateixa opció per a obtenir com a resultat el sumatori que s'ha d'avaluar per a obtenir el valor de l'aproximació fent

Exemple 21.3

```
> f:=x->x^2;
> RiemannSum(f(x),x=0..1,method=right,output=sum);
> RiemannSum(f(x),x=0..1,method=right,output=sum,partition=5);
```

Si repetiu l'exemple anterior per a la funció $\sin(x)$ entre 0 i 2π podeu observar que les sumes de Riemann aproximen l'àrea amb signe, és a dir, l'àrea sota l'eix d'abscisses és de signe negatiu. Recordeu també que per aconseguir aproximacions numèriques dels valors obtinguts podeu usar la comanda `evalf`.

Exercici 21.1

Construiu una seqüència que contingui les sumes de Riemann *avaluant en el costat dret* per a la funció $f(x) = x^2$ entre 0 i 1 partint l'interval en 2^n subintervalls, per a $n = 1$ fins a 15.

Si doneu un cop d'ull a l'ajuda de Maple sobre la comanda que estem estudiant veureu que l'opció `method=` pot tenir diferents valors. En particular podreu fer l'exercici següent.

Exercici 21.2

Repetiu l'exemple i l'exercici anteriors per a les sumes de Riemann *avaluant a l'esquerra* i *avaluant la funció en el punt mig de cada sub-interval*.

La integral de Riemann és el límit de les sumes de Riemann quan el nombre de subintervalls tendeix a infinit. En l'exemple següent aplicarem la definició per a calcular la integral de Riemann de la funció $f(x) = x^2$ entre 0 i 1.

Exemple 21.4

```
> f:=x->x^2;
> seqsup:=n-> RiemannSum(f(x),x=0..1,method=right,output=value,partition=n);
> intsup:=limit(seqsup(n),n=infinity);
```

Exercici 21.3

Calculeu el límit de l'exemple anterior però prenent les sumes per l'esquerra primer i les sumes obtingudes avaluant en el punt mig després. Observeu que s'obté el mateix resultat.

La comanda de Maple `int` (que està disponible sense carregar el paquet **Student**) permet obtenir el mateix resultat de forma automàtica. Comproveu-ho en l'exemple següent.

Exemple 21.5

```
> int(f(x), x=0..1);
```

Maple conté comandes que s'anomenen inerts. Aquestes comandes només ens representen per pantalla la funció que els hem demanat però no retornen el valor del resultat (ja heu vist que es pot obtenir el resultat d'una suma de Riemann com una expressió en forma de sumatori i no com el valor d'aquest sumatori). Donen una oportunitat a l'usuari de comprovar que no ha comès cap error quan ha escrit la comanda o per a poder guardar una expressió formal en lloc del resultat final d'un càlcul. La comanda inert associada a la integral de Riemann és `Int`.

Exemple 21.6

```
> Int(f(x), x=0..1);
> value(%);
```

A més, la comanda `RiemannSum` també es pot aplicar a una expressió del tipus `Int(f(x), x=a..b)`

```
> RiemannSum(Int(f(x), x=0..1), method=right, output=sum);
> value(%);
```

Com heu vist en l'exemple anterior, per fer que una comanda inert s'executi i ens retorni el valor del que li hem demanat cal aplicar-li la comanda `value`.

21.2 El mètode dels trapezis

Sigui f una funció contínua en l'interval $[a, b]$. Dividim l'interval en n parts iguals i en cadascun d'aquests subintervalls $[a + ih, a + (i + 1)h]$ aproximem la funció pel trapezi que s'obté al substituir la gràfica de la funció entre $f(a + ih)$ i $f(a + (i + 1)h)$ pel segment que uneix aquests dos punts. Recordeu que l'àrea d'un trapezi és la semisuma de les bases per l'alçada. En la nostra situació, si $h = (b - a)/n$, l'àrea és $\frac{h}{2}(f(a + ih) + f(a + (i + 1)h))$.

La comanda `ApproximateInt` del paquet **Student[Calculus1]** permet calcular i visualitzar aquest tipus de càlculs. Per a calcular el valor de l'aproximació per trapezis de l'àrea sota la gràfica de $f(x) = e^x$ en l'interval $[0, 1]$ prenent una partició en 10 sub-intervals d'igual longitud es pot fer el següent.

Exemple 21.7

```
> ApproximateInt(exp(x),x=0..1,method=trapezoid,partition=10,output=value);
> evalf(%);
```

Podeu observar que la sintaxi és la mateixa que s'utilitza en la comanda `RiemannSum` i, per tant, no us ha de costar descobrir (si és necessari mirant el que diu l'ajuda de Maple) com respondre la pregunta de l'exercici següent.

Exercici 21.4

Utilitzeu `ApproximateInt` per a fer un gràfic on es representi la funció $f(x) = \sin(x)$ en l'interval $[0, \pi]$ i l'aproximació per trapezis que s'obté dividint l'interval en 4 parts de la mateixa longitud. Feu que el color del gràfic de la funció i el dels segments de les aproximacions per trapezis siguin diferents.

Com en el cas de les aproximacions de Riemann també es pot mirar el que passa si anem dividint l'interval en parts cada cop més petites.

Exercici 21.5

Considereu una funció que retorni les sumes trapezoidals de $g(x) = e^x$ entre 0 i 1 en funció del nombre n de divisions que es facin i calculeu-ne el límit. Compareu-lo amb la integral de Riemann, què observeu?

21.3 Mètode de Simpson

Tot seguit anem a descriure un altre mètode per a aproximar la integral de Riemann. Dividim l'interval $[a, b]$ en un nombre parell n de parts iguals, $x_i = a + ih$ per $i = 0, \dots, n$ on $h = (b - a)/n$. Per a cada interval $[x_{2i-2}, x_{2i}]$ considerem una funció quadràtica $q(x) = \alpha x^2 + \beta x + \gamma$ que coincideix amb f en els punts $x_{2i-2}, x_{2i-1}, x_{2i}$.

Donats tres punts $[a, b1]$, $[a + h, b2]$ i $[a + 2h, b3]$, anem a veure com calcular els coeficients de la funció quadràtica $q(x) = \alpha x^2 + \beta x + \gamma$ tals que $f(a) = b1$, $f(a + h) = b2$ i $f(a + 2h) = b3$. I tot seguit, la corresponent integral de Riemann entre a i $a + 2h$. Per això utilitzarem les tècniques de la interpolació (recordeu d'una pràctica anterior que les comandes referents a la interpolació es troben al paquet **CurveFitting**).

Exemple 21.8

```

> with(CurveFitting):
> PolynomialInterpolation([[a,b1],[a+h,b2],[a+2*h,b3]],x);
> q:=unapply(%,x);
> int(q(x),x=a..a+2*h);
> simplify(%);

```

Exercici 21.6

Apliqueu les comandes de l'exemple anterior a la funció $f(x) = \sqrt{x}$ amb $a = 0$ i $h = 0.5$, $h = 0.05$ i $h = 0.005$. Dibuixeu en una mateixa gràfica la funció f i les corresponents aproximacions quadràtiques.

El mètode de Simpson consisteix en aproximar la integral mitjançant àrees del tipus que acabem de descriure, és a dir, donada f en un interval $[a, b]$ i un nombre parell n , aproximem l'àrea sota la gràfica per la suma

$$\sum_{i=1}^{n/2} \frac{h}{3} (f(x_{2i-2}) + 4f(x_{2i-1}) + f(x_{2i}))$$

Maple calcula aquesta suma directament amb la comanda `ApproximateInt` i l'opció `method=simpson`.

Exemple 21.9

```

> g:=x->sqrt(x);
> ApproximateInt(g(x),x=0..1,method=simpson, output=sum,partition=6);
> evalf(%);
> ApproximateInt(g(x),x=0..1,method=simpson, output=value,partition=10);
> evalf(%);

```

Exercici 21.7

Repetiu l'exercici final de l'apartat anterior (exercici 21.5) però ara amb el mètode de Simpson.

21.4 La funció integral

Donada f una funció contínua i $a \in \mathbb{R}$, podem definir una nova funció com $F(x) = \int_a^x f(t) dt$, és a dir, a cada valor de x li assignem la integral de Riemann entre a i x . Aquesta funció s'obté amb Maple si en el lloc on s'especifica el rang per a x d'una comanda `int` no especifiquem cap rang concret per a la variable i deixem només el nom d'aquesta variable.

```
> f:=x->cos(x);
> F:=int(f(t),t=a..x);
> FF:=int(f(x),x);
```

(Noteu que no surt exactament el mateix resultat.)

Per a valors de x propers a a , podem aproximar la integral de Riemann per l'àrea obtinguda amb el mètode de Simpson. Per exemple, si $f(x) = \cos(x)$

```
> f:= x-> cos(x);
> F:= x-> ApproximateInt(f(y),y=0..x,method=simpson,partition=2,output=value);
> evalf(F(Pi));
> evalf(F(Pi/2));
> evalf(F(Pi/4));
```

Recordeu que la derivada d'una funció F a $x = a$ és $\lim_{h \rightarrow 0} \frac{F(a+h) - F(a)}{h}$. Per tant, la derivada en $x = 0$ de $F(x)$ és

```
> limit(F(h)/h,x=0);
```

Què observeu? (compareu el resultat amb $f(0)$) Quina conclusió en treieu?.

21.5 Àrea entre dues gràfiques

Donades dues funcions contínues f i g , observeu que amb les eines que tenim també podem calcular l'àrea de la regió que delimiten.

Exercici 21.8

a) En una mateixa gràfica representeu les funcions $f(x) = 2x^2$ i $g(x) = x^3 - x + 2$ i sombregeu l'àrea que delimiten de color vermell (hi ha una opció de la comanda `plot` que permet donar un color determinat a la regió que queda entre el gràfic d'una funció i l'eix de les abscisses).

b) Calculeu l'àrea d'aquesta regió.

22 Càlcul de primitives

El Teorema Fonamental del Càlcul ens diu que tota funció integrable Riemann en un interval $[a, b]$ té una primitiva en aquest interval. Aquesta secció està dedicada al càlcul de primitives i es fa un breu repàs de les tècniques més habituals. Recordeu d'una pràctica anterior que si f és una expressió en x aleshores la comanda de Maple per obtenir la integral definida o de Riemann $\int_a^b f(x)dx$ és `int(f(x), x=a..b)`. Per a resoldre la integral indefinida o càlcul d'una primitiva s'utilitza `int(f(x), x)`.

Exemple 22.1

```
> i:=int(1/(1+x^2), x);
```

És fàcil comprovar que el resultat és correcte fent una derivada de la funció obtinguda.

```
> diff(i, x);  
> diff(i+C, x);
```

També coneixem la comanda `Int`. Recordeu que és una comanda inert que només reproduïx per pantalla el que li hem demanat (de manera que podem comprovar que no ens hem equivocat a l'introduir la funció) i que per a obtenir el càlcul explícit s'ha de combinar amb la comanda `value`.

Exemple 22.2

```
> Int(1/(1+x^2), x);  
> value(%);
```

Ara bé, no tota funció primitiva es pot descriure en termes de *funcions elementals*. Fixeu-vos que passa quan li demaneu a Maple que calculi la funció primitiva de e^{-x^2} .

Exemple 22.3

```
> Int(exp(-x^2), x); value(%);
```

Observeu que en el resultat surt una nova funció que Maple anomena `erf`. Utilitzeu l'ajuda de Maple per a trobar la definició de la funció `erf(x)` i calculeu la seva derivada.

```
> ?erf;  
> diff(erf(x), x);
```

De les funcions primitives que no es poden expressar com a combinació de funcions elementals se'n diu funcions el·líptiques. Maple els hi assigna un nom com podeu veure a l'exemple anterior. En el següent exemple veureu unes quantes primitives més d'aquest tipus.

Exemple 22.4

```
> int(sqrt(x^3+1),x);
> int(1/(1-3*x^2)/sqrt(1-x^2)/sqrt(1-4*x^2),x);
> int(sqrt(1-4*x^2)/sqrt(1-x^2),x);
```

Amb l'ajuda de Maple podeu veure els diferents tipus de funcions el·líptiques que té definides. Tot seguit anem a veure un altre tipus de situació en què ens podem trobar.

Exemple 22.5

```
> g:=Int((3*x^2+1)/sqrt((1-x-x^3)*(1+x+x^3)),x);
> value(g);
```

Malgrat que Maple no pot avaluar la integral anterior directament amb la comanda `int`, si que disposa de les tècniques estàndard d'integració en el paquet **Student[Calculus1]** per ajudar-nos (recordeu que abans d'utilitzar les comandes del paquet heu de fer `with(Student):`). Amb aquestes eines es pot simplificar l'expressió de la integral indefinida de manera que Maple pot finalment calcular-la. Analitzarem els mètodes de canvi de variable, integració per parts i estudiarem el cas en que la funció és un quocient de polinomis.

22.1 Canvi de variable o substitució

La integració per substitució o canvi de variable es basa en la regla de la cadena i és efectiva si tenim una integral expressable en la forma $\int f(g(x))g'(x)dx$. Aleshores, si fem la substitució $g(x) = u$ i per tant $g'(x)dx = du$, resultarà una nova integral indefinida de la forma $\int f(u)du$ que, si hem fet el canvi adequat, serà més fàcil de calcular.

Maple aplica un canvi de variable en una integral amb la comanda `Rule` del paquet **Student[Calculus1]**. La sintaxi d'aquesta comanda és `Rule[change,f(u)=h(x)](integral)` on `integral` és la integral en la variable `x`, `f(u)=h(x)` és la substitució que desitgem fer i `u` és la nova variable en la integral.

Exemple 22.6

```
> f:=Int((exp(x)-2*exp(2*x))/(1-exp(x)),x);
> g:=Rule[change,exp(x)=u](f);
> value(rhs(g));
```

Ara bé, ens queda la tasca de desfer el canvi de variable fet: escriure la solució en termes de la variable original. Per això el resultat de la comanda `Rule` ja està preparat per a fer la feina de substitució i donar el resultat final.

```
> value(g);
```

Exercici 22.1

Quin és el canvi més adient per a resoldre el problema de l'exemple 22.5?

Exercici 22.2

Simplifiqueu al màxim mitjançant un canvi de variable el càlcul de primitives per les següents funcions:

a) $\int \frac{x + 1 + 2\sqrt{x}}{x - \sqrt{x}} dx$

b) $\int \frac{e^x - 2e^{2x}}{1 - e^x} dx$

c) $\int \cos(x)^5 dx$ (recordeu la relació trigonomètrica $\sin(x)^2 + \cos(x)^2 = 1$).

22.2 Mètode d'integració per parts

El mètode d'integració per parts es desprèn de la regla de la derivació del producte de funcions. Usualment, s'escriu $\int u dv = uv - \int v du$. Una vegada optem per aquest mètode, el problema bàsic és decidir quina funció cal prendre com a u .

La comanda `Rule` de l'apartat anterior també està preparada per a poder fer servir el mètode de la integració per parts. Podeu veure en l'exemple següent com s'aplica a la integral $\int xe^{2x} dx$.

Exemple 22.7

```
> f:=Int(x*exp(2*x),x);
> Rule[parts,x,1/2*exp(2*x)](f);
> value(%);
```

Noteu que en les opcions entre claudàtors es posen les dues funcions u i v de la fórmula.

També és possible aplicar de forma encadenada la regla d'integració per parts com en l'exemple següent.

Exemple 22.8

Anem a calcular una primitiva per $(3x^2 - 2x + 7) \cos(x)$ utilitzant el mètode d'integració per parts. Prenem per a començar $u = 3x^2 - 2x + 7$ i $v = \sin(x)$ ($dv = \cos(x) dx$).

```
> i:=Int((3*x^2-2*x+7)*cos(x),x);
> Rule[parts,3*x^2-2*x+7,sin(x)](i);
```

Ara podeu veure que hi ha un factor comú $\sin(x)$ en la integral. Per a poder aplicar un altre cop la integració per parts canviarem l'expressió tenint en compte aquest factor comú amb l'opció `rewrite` de la comanda `Rule`

```
> Rule[rewrite,6*x*sin(x)-2*sin(x)=(6*x-2)*sin(x)](%);
```

Finalment tornarem a aplicar la regla d'integració per parts prenent $u = 6x - 2$ i $v = -\cos(x)$ ($dv = \sin(x) dx$)

```
> Rule[parts,6*x-2,cos(x)](%);
> value(%);
```

Exercici 22.3

Amb un canvi de variable o bé integrant per parts les vegades que sigui necessari, simplifiqueu i calculeu les següents integrals indefinides:

a) $\int \frac{dx}{\cos(x)}$

b) $\int x^n \ln(x) dx$

c) $\int x^n e^x dx$

d) $\int x \arctan(x) dx$

e) $\int \cos(x) e^x dx$

Exercici 22.4

Calculeu $\int x^2 \arctan(x) dx$. Si feu el canvi de variable $u = \arctan(x)$ haureu de resoldre una integral d'una funció racional, problema que comentarem a l'apartat següent.

22.3 Integrals racionals

Una funció racional és una funció que s'expressa com un quocient de polinomis. En aquest apartat analitzarem les diferents primitives que ens podem trobar a partir de funcions racionals així com el mètode per a calcular-les.

Abans de tot però anem a veure com, donada una expressió racional, tenim comandes que ens aïllen els polinomis del numerador i del denominador (`numer` i `denom`).

Exemple 22.9

```
> rac:=(x^2+3*x-6)/(x^5-7);
> p1:=numer(rac);
> p2:=denom(rac);
```

Anem a analitzar uns casos senzills de funcions racionals.

Exercici 22.5

- a) Fent un canvi de variable calculeu una primitiva per la funció $\frac{1}{3x-5}$. Quin és el canvi de variable més adient per calcular una integral del tipus $\frac{1}{ax+b}$?
- b) Calculeu una primitiva per $\frac{1}{(x-5)^2+9}$. En què es transforma si fem el canvi de variable $x-5=3\tan(u)$? Calculeu la primitiva, quina és la substitució que cal fer per a desfer el canvi?

El següent cas que analitzarem és el cas $\frac{1}{(ax^2+bx+c)^n}$. Aquest tipus d'integrals se simplifiquen si es reescriu l'expressió ax^2+bx+c de manera que sigui del tipus $a(x+d)^2+e$. D'aquesta tècnica se'n diu "completar quadrats". És fàcil comprovar que $d = \frac{b}{2a}$ i $e = \frac{-b^2}{4a} + c$. Aleshores la feina es podrà fer amb la funció que definireu quan feu l'exercici següent.

Exercici 22.6

Definiu un procediment que s'anomeni **completarquadrats**, tingui com argument una expressió quadràtica $p(x) = ax^2+bx+c$ i doni com a resultat la mateixa expressió vista com $p(x) = a\left(x + \frac{b}{2a}\right)^2 + \left(c - \frac{b^2}{4a}\right)$. Comproveu si el procediment funciona amb l'expressió $p(x) = 2x^2 - 6x + 18$.

Aleshores el canvi que cal fer (com ja hem vist a l'exercici anterior) és $x + d = \sqrt{e} \tan(u)$ i després utilitzar la relació trigonomètrica $1 + \tan(u)^2 = \frac{1}{\cos(u)^2}$ si cal.

Exemple 22.10

```
> r:=1/(x^2-6*x+18)^3;
> d:=denom(r);
> d1:=op(1,d);
> d2:=completarquadrats(d1);
> r1:=Int(r,x);
> r2:=Rule[rewrite,d1=d2](r1);
> r3:=Rule[change,x-3=3*tan(u)](r2);
```

Hem de forçar Maple a fer la darrera substitució $1 + \tan(u)^2 = \frac{1}{\cos(u)^2}$.

```
> r4:=Rule[rewrite,1+tan(u)^2=1/cos(u)^2](r3);
> r5:=value(rhs(r4));
> resultat:=value(r4);
> collect(resultat,atan(1/3*x-1));
```

Exercici 22.7

Calculeu una primitiva de la funció $\frac{3}{(2x^2 - 8x + 36)^2}$ amb el mètode anterior.

Anem a veure tot seguit què passa si al cas anterior afegim un polinomi de grau 1 al numerador.

Exercici 22.8

Calculeu una primitiva per la funció $\frac{2x + 2}{x^2 + 2x + 11}$ mitjançant un canvi de variable. Fixeu-vos que estem davant un cas particular de la situació general $\frac{2ax + b}{ax^2 + bx + c}$.

Ara bé, fixeu-vos que donada una funció racional del tipus $\frac{dx + e}{ax^2 + bx + c}$ es pot descompondre com a suma de dues dels tipus que ja hem estudiat:

$$\frac{\frac{c}{2a}(2ax + b)}{ax^2 + bx + c} + \frac{d - \frac{eb}{2a}}{ax^2 + bx + c}.$$

Finalment, observeu que tots els casos vistos fins ara corresponen a tipus de funcions racionals en que el grau del numerador és menor que el grau del denominador. En general i donada qualsevol funció racional sempre ens podem reduir a aquesta situació utilitzant la divisió de polinomis. Si $\frac{P}{Q}$ és una funció racional, la podem escriure com $\frac{P}{Q} = q + \frac{r}{Q}$ on q i r son el quocient i el reste de la divisió de polinomis. Aleshores el grau de r és menor que el grau de Q . Recordeu les comandes `quo` i `rem` per calcular el quocient i el reste de la divisió.

```
> rac1:=(4*x^3+8*x-7)/(x^2-6);
> num:=numer(rac1);
> den:=denom(rac1);
> q:=quo(num,den,x);
> r:=rem(num,den,x);
> rac2:=q+r/den;
```

Per acabar, tota funció racional en la que el grau del numerador és menor que el del denominador es pot escriure com a suma de funcions dels tipus estudiats al principi d'aquest apartat. D'aquesta descomposició se'n diu la descomposició en fraccions parcials. Maple calcula aquesta descomposició directament mitjançant `convert`.

Exemple 22.11

```
> convert(rac1,parfrac,x);
> w:=(x^4+2)/(x^3+x^2+x);
> convert(w,parfrac,x);
```

Observeu que el càlcul d'una primitiva per a w és immediat després de expressar-la com a suma de fraccions del tipus estudiat.

Exercici 22.9

Donades $f(x) = \frac{3x-7}{(x-1)^2(x-2)^3}$ i $g(x) = \frac{x^5+x+2}{x^3+x^2+x}$.

- Doneu la descomposició de f i g en suma de fraccions del tipus estudiat al principi de la secció.
- Per a cadascuna de les funcions racionals obtingudes a l'apartat anterior, digueu quin és el canvi (o canvis) necessaris per a resoldre-la.
- Calculeu una primitiva de f i una de g .

Part VIII

Diagonalització i formes de Jordan

23 Endomorfismes, vectors i valors propis, diagonalització...

Un dels problemes típics quan es treballa amb endomorfismes d'un espai vectorial de dimensió finita és el de determinar bases respecte de les quals la matriu de l'endomorfisme és el més simple possible. Per a poder fer aquest estudi hi juga un paper fonamental la determinació dels valors propis i la dels subespais de vectors propis associats. En el paquet **LinearAlgebra** de Maple es poden trobar funcions per a realitzar totes les operacions necessàries.

23.1 Polinomi característic i valors propis

Per a determinar el polinomi característic d'una matriu M podem realitzar els càlculs a partir de la definició: construint la matriu $M - x \text{Id}$ i demanant el determinat d'aquesta nova matriu.

```
> with(LinearAlgebra):  
> M:=< <1,1,3> | <2,7,-2> | <1,3,-6> >;  
> Mx:= simplify(M-x*IdentityMatrix(3));  
> de:=Determinant(Mx);
```

O bé demanant directament que Maple faci el càlcul i ens retorni directament el polinomi amb la comanda `CharacteristicPolynomial`

```
> CharacteristicPolynomial(M,x);
```

(triar una cosa o l'altre només és una qüestió de gust personal ja que els càlculs que s'han de realitzar internament són essencialment els mateixos).

Per a determinar els valors propis (i la seva multiplicitat) caldrà determinar les arrels del polinomi característic. També es pot fer directament amb la comanda `solve` o amb la comanda específica del paquet **LinearAlgebra** que fa aquesta feina (`Eigenvalues`).

```
> M1:=< <1,1,3> | <-1,7,-2> | <1,3,-6> >;  
> solve(Determinant(M1-x*IdentityMatrix(3)));  
> Eigenvalues(M1);
```

Com podreu veure, la diferència entre els dos resultats és ben poca.

23.2 Vectors propis

Un cop es coneixen els valors propis associats a una matriu A , la determinació dels espais de vectors \vec{v} per als que es verifica alguna de les condicions $A\vec{v} = \lambda\vec{v}$ es redueix a la solució de sistemes d'equacions lineals homogenis, $(A - \lambda)\vec{v} = 0$. Recordeu que la comanda `NullSpace` del paquet **LinearAlgebra** resol aquest tipus de sistemes.

Exemple 23.1

La matriu $A = \begin{pmatrix} -2 & 2 & -3 & -1 \\ 2 & -1 & 2 & 0 \\ 3 & -2 & 4 & 1 \\ -2 & 0 & -2 & -1 \end{pmatrix}$ té per valors propis 1 i -1 tal i com es pot comprovar amb

```
> A:= <<-2|2|-3|-1>, <2|-1|2|0>, <3|-2|4|1>, <-2|0|-2|-1>>;
> Eigenvalues(A);
```

Els espais de vectors propis amb valor propi 1 o -1 es podran determinar amb

```
> B1:= A-IdentityMatrix(4);
> E1:=NullSpace(B1);
> B2:= A+IdentityMatrix(4);
> E2:=NullSpace(B2);
```

Podreu veure que $E1$ i $E2$ contenen cada una un parell de vectors que generen l'espai de vectors propis corresponent (multipliqueu A per cada una de les columnes que surten en $E1$ o $E2$ per a comprovar que es tenen, efectivament, vectors propis amb valor propi 1 o -1). A la vista dels resultats es pot afirmar que la matriu A és diagonalitzable i es pot confirmar aquest fet considerant

```
> E:=<op(E1)|op(E2)>;
> E^(-1).A.E;
```

Tot el procés de calcular els vectors propis associats a cada un dels valors propis d'una matriu té una comanda específica en el paquet `LinearAlgebra` que es crida amb `Eigenvectors`. Si considereu

```
> EE:=Eigenvectors(A);
```

obtindreu una seqüència amb dos elements, el primer és una columna amb els valors propis (repetits tants cops com la multiplicitat) i el segon és una matriu que té per columnes les components dels vectors propis ordenades de la mateixa manera que els valors propis (si us hi fixeu, veureu que no surten exactament els mateixos vectors que quan apliqueu `NullSpace`). Ara bé, si feu la multiplicació

```
> EE[2]^(-1).A.EE[2];
```

tornareu obtenir la matriu diagonal equivalent a la matriu A .

Amb tot el que heu vist fins ara, l'única dificultat de l'exercici següent hauria de ser la introducció de la matriu M .

Exercici 23.1

Considereu la matriu

$$M = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -5 & -4 & 0 & -6 & 6 & 4 & -1 \\ 0 & 1 & 1 & -2 & 3 & 3 & -1 \\ -7/2 & -9/2 & -1/2 & -11/2 & 11/2 & 3 & -1/2 \\ -3/2 & -3/2 & 5/2 & -11/2 & 17/2 & 4 & -3/2 \end{pmatrix}$$

- Calculeu el polinomi característic de M .
- Determineu els seus valors propis i les multiplicitats corresponents.
- Determineu els espais de vectors propis associats a cada un dels valors propis.
- Comproveu que la matriu M és diagonalitzable.

e) Calculeu una matriu C tal que $C^{-1} \cdot M \cdot C$ sigui diagonal.

En aquest altre exercici podreu veure una matriu amb el mateix polinomi característic que la matriu M però que no és diagonalitzable.

Exercici 23.2

Considereu

$$B = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -4 & -3 & 1 & -5 & 6 & 3 & -1 \\ 1 & 2 & 2 & -1 & 3 & 2 & -1 \\ -5/2 & -7/2 & 1/2 & -9/2 & 11/2 & 2 & -1/2 \\ 1/2 & 1/2 & 9/2 & -7/2 & 17/2 & 2 & -3/2 \end{pmatrix}$$

calculeu el seu polinomi característic i comproveu que no és diagonalitzable.

Una de les propietats bàsiques del polinomi característic d'una matriu és que *anul·la* a la matriu corresponent. És a dir, si $p(x)$ és el polinomi característic de la matriu A es compleix $p(A) = 0$. En l'exercici següent es posa de manifest aquest fet en un parell d'exemples.

Exercici 23.3

Considereu les matrius A donades per

$$\begin{pmatrix} 1 & 3 & 4 & -1 \\ 0 & 3 & 2 & 1 \\ -1 & -1 & 2 & 5 \\ 2 & 3 & 1 & -1 \end{pmatrix} \quad \begin{pmatrix} -2 & 3 & 1 & -1 \\ 2 & -1 & 0 & 1 \\ -1 & -2 & 2 & 3 \\ 3 & 3 & 1 & -1 \end{pmatrix}$$

calculeu els seus polinomis característics i comproveu que les anul·len.

D'entre tots els polinomis que anul·len un endomorfisme s'en pot determinar el que té el grau més petit (polinomi mínim). La comanda de Maple `MinimalPolynomial` calcula el polinomi mínim d'una matriu.

Exemple 23.2

Es pot calcular el polinomi mínim de $A = \begin{pmatrix} 0 & -1 & 2 \\ 3 & -4 & 6 \\ 2 & -2 & 3 \end{pmatrix}$ amb

```
> A:= < <0,3,2>|<-1,-4,-2>|<2,6,3> >;  
> pm:=MinimalPolynomial(A,x);  
> Pm:=unapply(pm,x);  
> Pm(A);
```

Observeu que la funció Pm que hem definit s'aplica correctament sobre una matriu i sobre A dóna com a resultat la matriu nul·la.

Exercici 23.4

Calculeu el polinomi característic de la matriu A anterior i comproveu que té els mateixos factors irreductibles que el polinomi mínim (només que amb multiplicitats menors).

24 Formes de Jordan

Com ja haureu vist, un endomorfisme d'un espai vectorial (matriu quadrada) no sempre diagonalitza. Per a aquests casos és possible intentar obtenir altres *formes canòniques*. En l'exemple següent veureu una matriu 3×3 que té un únic valor propi i que, a més, l'espai de vectors propis només té dimensió 1.

Exercici 24.1

$$\text{Sigui } A = \begin{pmatrix} 1 & 0 & 1 \\ -1 & 3 & 1 \\ 0 & -1 & 2 \end{pmatrix}.$$

- Calculeu el polinomi característic de A , comproveu que A té un únic valor propi λ (triple) i que l'espai de vectors propis de A té dimensió 1.
- Sigui \vec{v}_3 un vector propi no nul de A . Determineu un vector \vec{v}_2 tal que $(A - \lambda \cdot \text{Id})\vec{v}_2 = \vec{v}_3$ i un vector \vec{v}_1 tal que $(A - \lambda \cdot \text{Id})\vec{v}_1 = \vec{v}_2$.
- Comproveu que $\vec{v}_1, \vec{v}_2, \vec{v}_3$ són independents.
- Calculeu la matriu $P^{-1} \cdot A \cdot P$ si P és la matriu que té per columnes les components dels vectors \vec{v}_i .

La matriu que heu obtingut en l'últim apartat de l'exercici anterior és el que normalment s'anomena un *bloc de Jordan* (un mateix valor en tots els elements de la diagonal i 1 just en els llocs que estan per sobre, o per sota, d'aquesta diagonal). En general, es poden buscar formes canòniques de la forma

$$\left(\begin{array}{c|c|c} J_1 & 0 & \cdots \\ \hline 0 & J_2 & \cdots \\ \hline \vdots & \vdots & \ddots \end{array} \right), \text{ on cada un dels } J_i \text{ és un bloc de Jordan.}$$

De la mateixa forma que per als valors i vectors propis, dins del paquet **LinearAlgebra** hi ha comandaments específics per a determinar la forma de Jordan d'un endomorfisme i per a obtenir el canvi de base que permet obtenir aquesta forma a partir de qualsevol altra forma equivalent. La comanda que fa les dues feines és `JordanForm`. Per a obtenir la forma de Jordan equivalent a una matriu A donada només cal fer `JordanForm(A)`; si es vol la matriu de canvi es pot modificar la comanda anterior i posar `JordanForm(A, output='Q')`; i si es volen les dues coses `JordanForm(A, output=['J', 'Q'])`; Veieu a continuació uns exemples.

Exemple 24.1

```
> A:=<<2, 2, 0, 0>>|<1/2, 3/2, -1/2, 1/2>>|<0, 0, 2, 0>>|<-1/2, -3/2, 1/2, 3/2>>;
> JordanForm(A);
> B:= JordanForm(A,output='Q');
> C:= JordanForm(A,output=['J','Q']);
> simplify(B^(-1).A.B);
> simplify(C[2]^(-1).A.C[2]);
```

Amb la comanda anterior es pot fer l'exercici següent:

Exercici 24.2

Considereu la matriu A donada per

$$\begin{pmatrix} 3 & 0 & 0 & 1 & 4 & -3 & 2 \\ -1 & 3 & 0 & -2 & -5 & 4 & -2 \\ 0 & 2 & 3 & -3 & -1 & 2 & 1 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ -2 & -2 & -1 & 0 & -8 & 6 & -7 \\ -1 & -2 & -1 & 1 & -6 & 5 & -5 \\ 2 & 0 & 1 & 1 & 9 & -6 & 7 \end{pmatrix}$$

i la matriu B

$$\begin{pmatrix} 2 & 1 & -1 & 1 & 4 & -3 & 3 \\ 0 & 3 & 0 & -2 & -4 & 3 & -2 \\ 0 & 3 & 2 & -3 & 0 & 1 & 2 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ 0 & -4 & 1 & 0 & -8 & 6 & -9 \\ 0 & -4 & 1 & 1 & -7 & 6 & -7 \\ 0 & 1 & 0 & 1 & 8 & -5 & 8 \end{pmatrix}$$

Existeix alguna matriu invertible P tal que $P^{-1} \cdot A \cdot P = B$? (Determineu les formes de Jordan de A i B).

24.1 Potències de matrius.

Recordeu que per a calcular la potència n -èsima d'una matriu M diagonalitzable, es pot calcular la seva forma diagonal D , obtenir

$$M = CDC^{-1},$$

de tal manera que

$$M^n = CD^nC^{-1},$$

i aplicar finalment que elevar una matriu diagonal D amb coeficients a la diagonal d_1, \dots, d_m a una potència n és considerar la matriu diagonal que té per coeficients a la diagonal $(d_1)^n, \dots, (d_m)^n$.

Exercici 24.3

Doneu l'expressió general de M^n per a $M = \begin{pmatrix} -7 & -4 & 6 \\ 3 & 1 & -3 \\ -9 & -4 & 8 \end{pmatrix}$.

Exercici 24.4

Feu un procediment que tingui com argument una matriu M , que retorni un missatge d'error si la matriu no és diagonalitzable i que retorni l'expressió general per a M^n en cas contrari. Comproveu el funcionament del procediment amb la matriu de l'exercici anterior.

Exercici 24.5

- a) Com calcularieu M^n si M no diagonalitza?
- b) Per exemple, com obtindrieu la fórmula general

$$\begin{pmatrix} 2 & 1 & 0 \\ 0 & 2 & 1 \\ 0 & 0 & 2 \end{pmatrix}^n = \begin{pmatrix} 2^n & n2^{n-1} & n(n-1)2^{n-3} \\ 0 & 2^n & n2^{n-1} \\ 0 & 0 & 2^n \end{pmatrix} ?$$